A NOVEL HANDS-ON APPROACH TO LEARNING FPGA DESIGN

Tuesday, December 6th, 2016

1:00 PM PST

UBM

EMBEDDED SYSTEMS CONFERENCE '16

#ESCsv

Pre-Requisites

You Will Need:

- **Laptop with Windows or Linux (***Tools not supported on Mac***)**
- **Installed Tools:**
  - Arduino IDE
  - Intel Quartus Prime Lite Edition
    - *Includes Modelsim-Intel FPGA Edition and Max 10 FPGA support*
- **A USB Mini cable for connecting XLR8 board to laptop**

Follow the instructions here:
http://www.aloriumtech.com/openxlr8/

Tutorial Download

- **LFSR Code Package:** ***https://github.com/AloriumTechnology/ESC-Tutorial***







- **Arduino Board Library URL:**

https://raw.githubusercontent.com/AloriumTechnology/Arduino_Boards/master/package_aloriumtech_index.json

XLR8 Development Platform



- Application Accelerator & Development Board
- Designed for Arduino Developer Community
- Based on Intel MAX 10 FPGA
- Programmable with Arduino IDE

FPGA Block Diagram



Xcelerator Blocks

An **Xcelerator Block (XB)** is an optimized hardware implementation of a specific processor-intensive function.

Custom hardware implemented on the same FPGA fabric

Tightly integrated with the microcontroller

XBs can access the same register space

Integrate with the instructions of the microcontroller

Available XBs

- Floating Point Math
- Servo Control
- NeoPixel Control
- Enhanced Analog-to-Digital Functionality
- Multiple SPI

XB Roadmap

- Proportional-Integral-Derivative (PID) control
- Event Counters and Timers
- Quadrature Encoders/Decoders
- Pulse Width Modulation (PWM)
- Multiple UARTS

Why use FPGA?



FASTER

HIGHER-PERFORMANCE

Overview of Tutorial

OpenXLR8

Methodology that allows XLR8 users to develop their own Xcelerator Blocks and upload them to the FPGA.



Module-Level Design and Simulation

- **Pseudorandom Number Generator**
  - Using a Linear Feedback Shift Register (LFSR)
  - 8-bit
  - 4-tap



Module-Level Design and Simulation



Integration into XLR8: LFSR Wrapper



Integration into XLR8: XLR8 Top Module



Design Synthesis



Upload to FPGA

XLR8
RPD





Run Sketch



lfsr_example | XLR8_LFSR.h

```
#include "XLR8_LFSR.h"

void setup() {
  Serial.begin(115200);
  XLR8_LFSR.set_seed(0x55);
  XLR8_LFSR.set_freerunning_mode(false);
}

void loop() {
  Serial.println(XLR8_LFSR.get_lfsr(), BIN);
  delay(1000);
}
```

Done Saving.

/dev/cu.usbs

```
11111101
11111011
11110111
11101110
11011100
10111000
1110001
11100011
11000111
10001110
11101
111011
1110110
11101101
11011010
10110100
1101000
11010001
10100011
1000111
10001111
11111
111111
1111110
11111100
11111001
11110011
11100110
11001101
10011011
110110
1101101
```

Autoscroll

Let’s Dive In!

Arduino IDE Setup

- Go to Sketch -> Include Library -> Manage Libraries…
- Search for “XLR8” and install XLR8Core and XLR8BuildTemplate
- Go to Tools -> Board -> Boards Manager…
- Search for “XLR8” and install Alorium XLR8 Boards



Project Directory Setup

- In your operating system's file explorer, go into your Documents directory, then Arduino/libraries
- Copy the "XLR8BuildTemplate" directory you just downloaded to a new directory named "XLR8Build"
- This will be where we build our project



But for today at ESC…

- Copy downloaded XLR8Build directory to Arduino libraries directory



Linear Feedback Shift Register
(LFSR)




```
assign feedback = ~(lfsr_data[7] ^ lfsr_data[5] ^ lfsr_data[4] ^ lfsr_data[3]);
```

RTL for the LFSR

- RTL = Register-Transfer Level
  - HDL code
  - Verilog/SystemVerilog
  - VHDL
- Located in XLR8Build/extras/rtl
- The LFSR module, alorium_lfsr.v

```
module alorium_lfsr
  (
   // Clock and Reset
   input clk,
   input reset_n,
   // Inputs
   input new_seed,
   input enable,
   input wire [7:0] seed,
   // Output
   output reg [7:0] lfsr_data
  );

   wire feedback;

   assign feedback = ~(lfsr_data[7] ^ lfsr_data[5] ^ lfsr_data[4] ^ lfsr_data[3]);

   always @(posedge clk or negedge reset_n) begin

      if (!reset_n) begin
         lfsr_data <= 8'h01 ;  // LFSR register cannot be all 1's for XNOR LFSR
      end
      else if (new_seed) begin
         lfsr_data <= &seed ? 8'h01 : seed ;  // LFSR register cannot be all 1's f
      end
      else if (enable) begin
         lfsr_data <= {lfsr_data[6:0],feedback};
      end // else: !if(!reset_n)
   end // always @ (posedge clk or negedge reset_n)

endmodule // alorium_lfsr
```

Testbench

- The testbench, alorium_lfsr_tb.v

```
include "alorium_lfsr.v"

odule alorium_lfsr_tb();

  reg clock, reset, new_seed, enable;
  reg [7:0] in;
  wire [7:0] out;

  initial begin
    clock = 1;
    reset = 1;
    new_seed = 0;
    enable = 0;
    #5 reset = 0;
    #10 reset = 1;
    #10 in = 8'b10101010;
    #15 new_seed = 1;
    #5 new_seed = 0;
    #5 enable = 1;
    #5 enable = 0;
    #25 enable = 1;
    #5 enable = 0;
    #25 enable = 1;
    #100;
    #5 $stop;
  end

  always begin
    #5 clock = ~clock;
  end

  alorium_lfsr lfsr_inst (
    // Clock and Reset
    .clk       (clock),
    .reset_n   (reset),
    // Inputs
    .new_seed  (new_seed),
    .enable    (enable),
    .seed      (in),
    // Output
    .lfsr_data (out));

ndmodule
```

Simulating the Testbench

- Start Modelsim
- File -> New -> Library…
- Create the default “work” library inside of our project RTL directory
- Compile -> Compile…
- Select alorium_lfsr.v and alorium_lfsr_tb.v
- “Compile” and then “Done”
- Open the testbench in the work area



Simulating the Testbench Continued

- Select our testbench signals and bring them into a waves window
- Hit the “Run –all” button



XLR8 Module

- xlr8_lfsr.v
- Connects the signals from the XLR8 core to the LFSR module
- Instantiates the alorium_lfsr module
- Controls register access

```
assign ctrl_sel = (dm_sel && ramadr == LFSR_CTRL_ADDR);
assign ctrl_we  = ctrl_sel && (ramwe);
assign ctrl_re  = ctrl_sel && (ramre);
assign seed_sel = (dm_sel && ramadr == LFSR_SEED_ADDR);
assign seed_we  = seed_sel && (ramwe);
assign seed_re  = seed_sel && (ramre);
assign data_sel = (dm_sel && ramadr == LFSR_DATA_ADDR);
assign data_we  = data_sel && (ramwe);
assign data_re  = data_sel && (ramre);
assign dbus_out =  ({8{ctrl_sel}} & lfsr_ctrl) |
                   ({8{seed_sel}} & lfsr_seed) |
                   ({8{data_sel}} & lfsr_data);
assign io_out_en = ctrl_re ||
                   seed_re ||
                   data_re;

always @(posedge clk or negedge rstn) begin
   if (!rstn)  begin
      lfsr_ctrl <= {WIDTH{1'b0}};
   end else if (clken && ctrl_we) begin
      lfsr_ctrl <= dbus_in[WIDTH-1:0];
   end
end // always @ (posedge clk or negedge rstn)

always @(posedge clk or negedge rstn) begin
   if (!rstn)  begin
      lfsr_seed <= {WIDTH{1'b0}};
   end else if (clken && seed_we) begin
      lfsr_seed <= dbus_in[WIDTH-1:0];
   end
end // always @ (posedge clk or negedge rstn)

alorium_lfsr lfsr_inst (
                    // Clock and Reset
                    .clk       (clk),
                    .reset_n   (rstn),
                    // Inputs
                    .new_seed  (seed_we),
                    .enable    (lfsr_ctrl[0] | data_re),
                    .seed      (lfsr_seed),
                    // Output
                    .lfsr_data (lfsr_data));
```

XB Addresses

- xb_adr_pack.vh
- Declare the address locations of your registers
- Refer to the XLR8 User Manual to find open register space

```
// ************************************************
// AVR address constants (localparams)
//  for registers used by Xcelerator Blocks (XBs)
// ************************************************

localparam LFSR_CTRL_Address = 8'he0;
localparam LFSR_SEED_Address = 8'he1;
localparam LFSR_DATA_Address = 8'he2;
```

LFSR Register Descriptions

LFSR Control								Address 0xE0
Bit	7	6	5	4	3	2	1	0
Function	Unused							Freerunning Mode
R/W	R	R	R	R	R	R	R	R/W
Initial	0	0	0	0	0	0	0	0

LFSR Seed								Address 0xE1
Bit	7	6	5	4	3	2	1	0
Function	LFSR Seed Data							
R/W	R/W	R/W	R/W	R/W	R/W	R/W	R/W	R/W
Initial	0	0	0	0	0	0	0	0

LFSR Data								Address 0xE2
Bit	7	6	5	4	3	2	1	0
Function	LFSR Result Data							
R/W	R	R	R	R	R	R	R	R
Initial	0	0	0	0	0	0	0	0

XLR8 Top

- xlr8_top.v
- Instantiate the xlr8_lfsr module
- Add the control signals to “stgi_xf_io_slv_dbusout” and “stgi_xf_io_slv_out_en”
- Don’t forget to declare your control signals

```
assign stgi_xf_io_slv_dbusout = xlr8_clocks_out_en      ? xlr8_clocks_dbusout :
                                xlr8_lfsr_slv_out_en    ? xlr8_lfsr_slv_dbusout :
                                                          xlr8_gpio_dbusout;
assign stgi_xf_io_slv_out_en  = xlr8_clocks_out_en ||
                                xlr8_lfsr_slv_out_en ||
                                xlr8_gpio_out_en;


 xlr8_lfsr #(
             .LFSR_CTRL_ADDR (LFSR_CTRL_Address),
             .LFSR_SEED_ADDR (LFSR_SEED_Address),
             .LFSR_DATA_ADDR (LFSR_DATA_Address),
             .WIDTH          (8)
            )
 lfsr_inst  (
             // Clock and Reset
             .rstn        (core_rstn),
             .clk         (clk_io),
             .clken       (1'b1),
             // I/O
             .dbus_in     (io_arb_mux_dbusout),
             .dbus_out    (xlr8_lfsr_slv_dbusout),
             .io_out_en   (xlr8_lfsr_slv_out_en),
             // DM
             .ramadr      (core_ramadr_lo8[7:0]),
             .ramre       (core_ramre),
             .ramwe       (core_ramwe),
             .dm_sel      (core_dm_sel)
            );

endmodule
```

Modify the Project QSF File

- xlr8_top.qsf under the “quartus” directory
- Add in our module files and the register address file

```
:) 2016 Alorim Technology.  All right reserved.

:ings for XLR8 project
.aloriumtech.com/xlr8
:hub.com/AloriumTechnology


../XLR8Core/extras/quartus/xlr8_top_core.qsf

signment -name QXP_FILE ../../../XLR8Core/extras/quartus/xlr8_atme


ssignment -name VERILOG_FILE ../../../XLR8ExampleXB/extras/rtl/xlr
signment -name VERILOG_FILE ../../../XLR8Build/extras/rtl/alorium_
signment -name VERILOG_FILE ../../../XLR8Build/extras/rtl/xlr8_lfs
signment -name VERILOG_FILE ../../../XLR8Build/extras/rtl/xb_adr_p

:l, etc.
signment -name SYSTEMVERILOG_FILE ../rtl/xlr8_top.v
signment -name TOP_LEVEL_ENTITY xlr8_top
signment -name SDC_FILE ../../../XLR8Core/extras/quartus/xlr8_top.

signment -name FLOW_ENABLE_POWER_ANALYZER OFF
signment -name EDA_SIMULATION_TOOL "ModelSim-Altera (Verilog)"
signment -name EDA_TIME_SCALE "1 ps" -section_id eda_simulation
```

Compile the Project in Quartus

- Open Quartus and open our project QSF file with File -> Open Project…
- Begin the compile with Processing -> Start Compilation
- After compilation is completed, File -> Convert Programming Files…
- Open Conversion Setup Data, open “openxlr8.cof,” and Generate



Burn the FPGA Image

- Open the Arduino IDE
- Under Tools -> Board select OpenXLR8
- Connect your board via USB and make sure it is selected in Arduino under Tools -> Port
- Tools -> Burn Bootloader



Arduino Library for the LFSR

- XLR8_LFSR.h
- Defines the same register addresses as in the RTL
- Sets and reads the LFSR registers

```
#ifndef _XLR8_LFSR_H_INCLUDED
#define _XLR8_LFSR_H_INCLUDED

#include <Arduino.h>

#define XLR8_LFSR_CTRL _SFR_MEM8(0xE0)
#define XLR8_LFSR_SEED _SFR_MEM8(0xE1)
#define XLR8_LFSR_DATA _SFR_MEM8(0xE2)

class XLR8_LFSRClass {
public:
  XLR8_LFSRClass() {}
  ~XLR8_LFSRClass() {}
  void set_seed(uint8_t seed) {
    XLR8_LFSR_SEED = seed;
  }
  uint8_t get_lfsr() {
    return XLR8_LFSR_DATA;
  }
  void set_freerunning_mode(boolean freerunning) {
    XLR8_LFSR_CTRL = freerunning;
  }
private:
};

extern XLR8_LFSRClass XLR8_LFSR;

#endif
```

Arduino LFSR Example

- Include the XLR8_LFSR.h
- Set the seed, enter a loop to print the result of the LFSR to serial output
- Compile and run on the board



Thank You!

Questions?

UBM

EMBEDDED SYSTEMS CONFERENCE '16

@ESC_Conf
#ESCsv